3 Gebundene Bewegungen

3.2 Das einfache Pendel



Fig.3.2-1

Ein weiteres Beispiel einer gebundenen Bewegung ist ein einfaches Pendel ("mathematisches" Pendel), das aus einer punktförmigen Masse m besteht, die an einem "starren" Faden der Länge l aufgehängt ist. Die Masse des Fadens ist gegenüber der Punktmasse verschwindend klein. Die Punktmasse steht mit zwei Körpern in WW, mit der Erde, $m\vec{g}$, und mit dem Faden, $\vec{T}$. Die Fadenspannung ist die Bindungskraft (Zwangskraft), die die Punktmasse auf einem Kreis in der x-y-Ebene hält. Zu Beginn wird die Masse seitlich bis Punkt A ausgelenkt (Winkel α_0). Nach dem Loslassen schwingt m bis zum Punkt A' und kehrt wieder nach A zurück usw.

Wir zerlegen die Kraft $m\vec{g}$ in eine Radialkomponente $\vec{F}_r = mg\cos(\alpha)\vec{r}^0$ in Richtung des radialen Einheisvektors $\vec{r}^0$ und in eine tangentiale Komponente $\vec{F}_t = -mg\sin(\alpha)\vec{t}^0$ in Richtung des Tangenteneinheitsvektors $\vec{t}^0$. Das negative Vorzeichen erscheint in $\vec{F}_t$, weil diese Komponente immer im Gegensinn der Auslenkung $\overset{\frown}{s} = \overset{\frown}{OA}$ wirkt. Der Bogen $\overset{\frown}{s}$ hat die Länge s. (Der Winkel α im Bogenmaß ist definiert als $\alpha := s/r$. Der Bruch $\Delta\alpha/\Delta t$ ist die Winkelgeschwindigkeit ω. Zwischen dieser Geschwindigkeit und der Bahngeschwindigkeit besteht die Beziehung $v = r\omega$. Denn $v = \Delta s/\Delta t = r\ \Delta\alpha/\Delta t = r\omega$, vgl. 3.4.1)

Im Falle der Kreisbewegung ist die Tangentialbeschleunigung gleich

$$a_t = r\ddot{\alpha} = l\frac{d^2\alpha}{dt^2} \tag{3.2-1}$$

Die Normal-oder Zentripetalbeschleunigung ist gegeben durch

$$a_n = v^2/r = v^2/l \tag{3.2-2}$$

Die Bewegungsgleichung des mathematischen Pendels lautet demnach

$$ma_t = ml\ddot{\alpha} = -mg\sin(\alpha)$$

Vereinfacht erhalten wir

$$\ddot{\alpha} = -g\sin(\alpha)/l \tag{3.2-3}$$

Wegen des $\sin(\alpha)$-Faktors ist es nicht möglich, diese Gleichung analytisch zu lösen. Wenn der Winkel aber klein ist (z.B. $\alpha < 5°$), können wir näherungsweise $\sin(\alpha)$ durch α selbst ersetzen (im Bogenmaß). Für derartig kleine Schwingungen reduziert sich (3.2-3) auf

$$\ddot{\alpha} + \frac{g}{l}\alpha = 0 \tag{3.2-4}$$

Wir werden diese Gleichung in 3.2.1 weiter untersuchen, vorher wollen wir aber Gleichung (3.2-3) numerisch mit Hilfe von `numeric::odesolve2` von MuPAD lösen. Wir greifen auf das Programm von 2.5 zurück und kürzen es, da wir es jetzt ja nur mit *einer* Differentialgleichung zweiten Grades zu tun haben. Wir erstellen zunächst einen Graphen für α in Funktion von t.

```
reset():
g:=9.8: l:=2://Laenge = 2m
a0:=45*PI/180://a = urspruengliche Auslenkung 45 Grad
IVP:={alpha''(t)+g/l*sin(alpha)=0,alpha'(0)=0,
alpha(0)=a0}:
fields:=[alpha(t),alpha'(t)]:
ivp:=numeric::ode2vectorfield(IVP, fields):
Y := numeric::odesolve2(ivp):
/*float(Y(1)[1]*180/PI);Y(1)[2]*l;
-24.04521625 // alpha in Grad nach 1 s
-2.842499313 // Geschwindigkeit nach 1s (m/s)*/
//Animation
dt:=0.05:imax:=100:
plot(
plot::Point2d(t,Y(t)[1]*180/PI, Color = RGB::Blue,
VisibleFromTo = t..t + 0.99*dt,
PointSize = 2*unit::mm
) $ t in [i*dt $ i = 0..imax],
plot::Line2d([t-dt,Y(t - dt)[1]*180/PI],
[t,Y(t)[1]*180/PI], Color = RGB::Red,
VisibleAfter = t)
$ t in [i*dt $ i = 1..imax])
```



Fig.3.2-2

Mit der Zeile `float(Y(1)[1]*180/PI);Y(1)[2]*l` fragen wir Winkel und Geschwindigkeit nach 1 Sekunde ab: Nach 1 Sekunde ist das Pendel zum ersten Mal auf der linken Seite der Senkrechten OB. Sein Ausschlag beträgt -24.045... Grad und seine Geschwindigkeit beträgt -2.842...m/s. (Beachte, dass Y(1)[2] die Winkelgeschwindigkeit ω nach 1 Sekunde ist. Die Bahngeschwindigkeit beträgt in diesem Augenblick $v = l\omega$.)

Die x-Achse zeigt die Zeit an und die y-Achse den Winkel α. Die Schwingungsdauer (Periode T) beträgt etwa 3 Sekunden. Um es genauer zu wissen, müssen wir den Zeitunterschied zwischen zwei Punkten bestimmen, in denen sich die Bewegung wiederholt, d.h. wir berechnen die Zeiten von zwei benachbarten, gleichphasigen Nullstellen von $\alpha(t)$, z.B. die beiden Nulldurchgänge in den Intervallen [0,1] und [3,4]. Die Differenz dieser Zeiten ist die Periodendauer T.

```
t2:=numeric::fsolve(Y(t)[1]=0,t=3..4);
t1:=numeric::fsolve(Y(t)[1]=0,t=0..1);
Periode:=t2[1]-t1[1]
[t = 3.689895347]
[t = 0.7379790694]
0 = 2.951916278
```

Die Periodendauer ist also $T = 2.952\ s$.

Wenn wir sehen wollen, wie das Pendel wirklich schwingt, so haben wir die Zeilen der Animation abzuändern

```
reset():
g:=9.8: l:=2://Laenge = 2m
a0:=45*PI/180://a = urspruengliche Auslenkung 45 Grad
IVP:={alpha''(t)+g/l*sin(alpha)=0,alpha'(0)=0,
alpha(0)=a0}:
fields:=[alpha(t),alpha'(t)]:
ivp:=numeric::ode2vectorfield(IVP, fields):
Y := numeric::odesolve2(ivp):
```

```
  //Animation
dt:=0.05:imax:=100:
plot(
plot::Point2d(sin(Y(t)[1])*l,l*(1-cos(Y(t)[1])), Color = RGB::Red,
VisibleFromTo = t..t + 0.99*dt,
PointSize = 3*unit::mm
) $ t in [i*dt $ i = 0..imax],
plot::Line2d([l*sin(Y(t)[1]),l*(1-cos(Y(t)[1]))],
[0,l], Color = RGB::Blue,
VisibleFromTo = t..t + 0.99*dt)
$ t in [i*dt $ i = 1..imax])
```



Fig.3.2-3

Wir haben die x-y-Koordinaten des schwingenden Massepunktes in die Animation eingeführt.



Fig.3.2-4

Wir sehen aus Fig.3.2-4, dass

$$x = l\sin(\alpha); \quad (l-y)/l = \cos(\alpha) \text{ oder } y = l(1-\cos(\alpha))$$

Als **Übung** sollten Sie das Programm zu Fig. 3.2-2 so abändern, dass auch die Geschwindigkeit dargestellt wird. Sie haben nur den Animationsteil abzuändern, um die folgende Figur 3.2-5 zu erhalten:



Fig.3.2-5

Die grüne Kurve stellt die 20-fach vergrößerte Geschwindigkeit dar. Sie beginnt in (0,0), wird sofort negativ und hat ihre Extrema dann, wenn $\alpha = 0$ ist.

Zur Kontrolle geben ich Ihnen hier noch das Programm zu Fig.3.2-5

```
//Animation
dt:=0.05:imax:=100:
 plot(
plot::Point2d(t,Y(t)[1]*180/PI, Color = RGB::Blue,
VisibleFromTo = t..t + 0.99*dt,
PointSize = 2*unit::mm
) $ t in [i*dt $ i = 0..imax],
plot::Line2d([t-dt,Y(t - dt)[1]*180/PI],
[t,Y(t)[1]*180/PI], Color = RGB::Red,
VisibleAfter = t)
$ t in [i*dt $ i = 1..imax],
plot::Point2d(t,Y(t)[2]*20, Color = RGB::Red,
VisibleFromTo = t..t + 0.99*dt,
PointSize = 2*unit::mm
) $ t in [i*dt $ i = 0..imax],
plot::Line2d([t-dt,Y(t - dt)[2]*20],
[t,Y(t)[2]*20], Color = RGB::Green,
VisibleAfter = t)
$ t in [i*dt $ i = 1..imax])
```

3.2.1 Das vereinfachte mathematische Pendel

Kehren wir zurück zur sinusfreien Gleichung (3.2-4), in der wir noch die Bogenlänge $s = \alpha l$ einführen:

$$\frac{d^2 s}{dt^2} = -\frac{g}{l} s \tag{3.2-5}$$

Die allgemeine Lösung dieser Differentialgleichung ist von der Form

$$s = B\cos(\omega t) + C\sin(\omega t)$$

B und C sind Konstanten, die von Zusatzbedingungen abhängen. Für den Gebrauch ist es angenehmer, die Lösung in der folgenden mathematisch äquivalenten Form zu schreiben

$$s = A\sin(\omega t + \phi) \tag{3.2-6}$$

Die Amplitude A ist gegeben durch $A = \sqrt{s_0^2 + v_0^2/\omega^2}$und die Anfangsphase ϕ ist durch die Gleichungen $\sin(\phi) = s_0/A$ und $\cos(\phi) = v_0/(\omega A)$ gegeben. Die Größe $\omega t + \phi$ heißt einfach Phase. Die Sinusfunktion wiederholt sich immer dann, wenn die Phase um 2π wächst:

$$\sin(\omega t + \phi) = \sin(\omega t + \phi + 2\pi) = \sin(\omega(t + 2\pi/\omega) + \phi)$$

Also wiederholt sich die Bewegung des Massepunktes nach $T = 2\pi/\omega$ Sekunden. T ist die Periode der einfachen (harmonischen) Schwingungsbewegung. Die Frequenz der harmonischen Bewegung ist gleich der Anzahl der vollen Schwingungen pro Zeiteinheit (in der Sekunde), also $f = 1/T$. Wie bei der Kreisbewegung gilt die Gleichung $\omega = 2\pi f$. Die Größe ω heißt Kreisfrequenz. Setzt man (3.2-6) in (3.2-5) ein, so erhält man

$$T = 2\pi\sqrt{\frac{l}{g}} \tag{3.2-7}$$

T hängt also nur von l und g ab und ist unabhängig von der Masse des Pendels. Diese Tatsache ist eine Folge der Proportionalität von träger-und schwerer Masse, also von m_t und m_s, vgl. 2.1.4. Wenn diese Proportionalität nicht bestünde, hätten wir statt (3.2-7) die folgende Gleichung

$$T = 2\pi\sqrt{\frac{lm_t}{gm_s}}\ ,$$

d.h. Pendel aus verschiedenen Materialien müssten verschiedene Periodendauern haben. Aber schon Newton und Bessel haben gezeigt, dass dies nicht der Fall ist, und dass somit m_t und m_s einander proportional sind.

Wichtig bei unseren Überlegungen war, dass die Schwingungsweite (Amplitude) recht klein ist ($< 5°$). Für ein Pendel mit größerer Amplitude kann man folgende Näherung benutzen

$$T = 2\pi\sqrt{\frac{l}{g}}(1 + \frac{1}{4}\sin^2(\frac{\alpha}{2}) + \frac{9}{64}\sin^4(\frac{\alpha}{2}) + ...) \qquad (3.2\text{-}8)$$

Wir werden im nächsten Abschnitt darauf zurückkommen.

3.2.2 Mit Bleistift und Papier

Wir schauen uns nochmals in leicht veränderter Gestalt Gleichung (3.2-3) an

$$\frac{d^2s}{dt^2} = -g\sin(\frac{s}{l}) \qquad (3.2\text{-}9)$$

Die folgenden Modifikationen werden uns weiterbringen

$$\frac{d^2s}{dt^2} = \frac{dv}{dt} = \frac{dv}{ds}\frac{ds}{dt} = \frac{dv}{ds}v = \frac{d(v^2/2)}{ds},$$

denn wir gelangen zu einer Gleichung, die sich integrieren lässt

$$\frac{d(\frac{1}{2}v^2)}{ds} = -g\sin(\frac{s}{l}).$$

Wir integrieren und dividieren durch die Masse

$$\tfrac{m}{2}v^2 - mgl\cos(\frac{s}{l}) = \tfrac{m}{2}v_0^2 - mgl\cos(\frac{s_0}{l}) = const.$$

Dies bedeutet: während der Massepunkt hin und her schwingt, bleibt die Größe (3.2.-10) konstant

$$\frac{m}{2}v^2 - mgl\cos(\frac{s}{l}) \qquad (3.2\text{-}10)$$

Ihr Wert ist gleich $\tfrac{m}{2}v_0^2 - mgl\cos(\frac{s_0}{l})$. Die Anfangsgeschwindigkeit v_0 war bisher immer null. Den Anfangsbogen s_0 können wir ersetzen durch den in Radiant gemessenen Anfangswinkel $\alpha_0 = \frac{s_0}{l}$. Es ist immer von größter Bedeutung, eine Größe zu finden, die sich während eines Prozesses nicht ändert. Was wir gefunden haben, ist in Wirklichkeit ein neues Beispiel für den Energieerhaltungssatz. Wir schauen uns nochmals Fig. 3.2-4 und die Beziehungen

$$x = l\sin(\alpha); \quad (l-y)/l = \cos(\alpha) \text{ oder } y = l(1-\cos(\alpha))$$

an, um die übliche Form des Energiesatzes zu gewinnen. Zunächst erhalten wir

$$\tfrac{m}{2}v^2 - mg(l-y) = mv_0^2 - mg(l-y_0) = const.$$

und daraus

$$\frac{m}{2}v^2 + mgy = \frac{m}{2}v_0^2 + mgy_0 := E \tag{3.2-11}$$

Damit haben wir wieder das Ergebnis aus 1.1.2 erhalten, dass nämlich in jedem Bahnpunkt die Summe aus kinetischer- und potentieller Energie konstant ist, vgl. Gl. (1.1-7). Der Wert E der Gesamtenergie (oft W geschrieben) hängt von der WW ab und von den Anfangsbedingungen. Aus (3.2-10) erhalten wir folgende Gleichung zur Berechnung der Geschwindigkeit für beliebige Winkel

$$v^2 = v_0^2 + 2gl(\cos(\alpha) - \cos(\alpha_0)) \tag{3.2-12}$$

Leider können wir mit dieser Gleichung keine Aussage über das Vorzeichen der Geschwindigkeit machen, wir können also nicht sagen, ob das Pendel gerade nach links oder nach rechts schwingt. Auch ist es nicht möglich, die Gleichung $v = \frac{ds}{dt}$ geschlossen zu integrieren, denn man kann zeigen, dass das Integral nicht mit Hilfe elementarer Funktionen darstellbar ist, da es sich um ein elliptisches Integral handelt. Bekannt ist z.B. die Berechnung der Periode $T = 4K\sqrt{\frac{l}{g}}$mit Hilfe des folgenden elliptischen Integrals

$$K(k) = \int_0^{\pi/2} \frac{dt}{\sqrt{1 - k^2 \sin^2(t)}} \tag{3.2-13}$$

$k := \sin(\alpha/2)$ ist ein Parameter. Wir wollen dieses Integral numerisch mit MuPAD integrieren und uns eine Tabelle in Schritten von 5 Grad ausdrucken lassen (von 0 bis 45 Grad). Wir schreiben dazu eine Prozedur, die die Werte von α_0 und α_{end} als Eingabedaten übernimmt.

```
reset():// Berechnung der Funktion (3.2-13)
periode:=proc(a0,aend)
begin
g:=9.8:
l:=2:
DIGITS:=10:
print("Grad     K        T/s "):
for a from a0 to aend step 5 do
k:=sin(a*PI/360):
K:=numeric::int((1/sqrt(1-k^2*sin(x)^2)),x=0..PI/2):
T:= K*sqrt(l/g)*4:
print(a,K,T):
end_for:
end_proc:
periode(0,45)
```

```
"Grad     K        T/s "
0, 1.570796327, 2.83845379
5, 1.571544297, 2.839805384
10, 1.573792131, 2.843867256
15, 1.577551661, 2.850660785
20, 1.582842804, 2.860221966
25, 1.589693871, 2.872601951
30, 1.598142002, 2.887867858
35, 1.608233762, 2.906103827
40, 1.620025899, 2.927412406
45, 1.633586307, 2.951916278
```